

# Digital simulation of CV: 1D case

Excerpt from Prof. Kenji Kano's book on basic electrochemisrty

Modified on February 6, 2009 10:19 pm

## 1 Diffusion equation

The Fick's first and second law for diffusion and the differential equation for diffusion are

$$\vec{\mathbf{J}}(\mathbf{r}, t) = -D\nabla c(\mathbf{r}, t) \quad (1)$$

$$\frac{\partial c}{\partial t} + \text{div} \cdot \vec{\mathbf{J}}(\mathbf{r}, t) = 0 \quad (2)$$

$$\frac{\partial c}{\partial t} = D\nabla^2 C \quad (3)$$




Figure 1: Discrete model of diffusion

If we consider one-dimensional diffusion, and the above equations are approximated by discretized model for numerical simulation

$$J_k(t) = -D \frac{c_{k+1}(t) - c_k(t)}{\Delta z} \quad (4)$$

$$\begin{aligned} \frac{c_k(t + \Delta t) - c_k(t)}{\Delta t} \Delta x \Delta y \Delta z &= [J_{k-1}(t) - J_k(t)] \Delta x \Delta y \\ \frac{c_k(t + \Delta t) - c_k(t)}{\Delta t} &= \frac{J_{k-1}(t) - J_k(t)}{\Delta z} \\ &= \frac{D}{(\Delta z)^2} [c_{k+1}(t) - 2c_k(t) + c_{k-1}(t)] \end{aligned} \quad (5)$$

$$c_k(t + \Delta t) = c_k(t) + D_M [c_{k+1}(t) - 2c_k(t) + c_{k-1}(t)] \quad (6)$$

$$D_M = \frac{D\Delta t}{(\Delta z)^2} \quad (7)$$

$$c_k(t + \Delta t) = \begin{pmatrix} D_M & 1 - 2D_M & D_M \end{pmatrix} \begin{pmatrix} c_{k+1}(t) \\ c_k(t) \\ c_{k-1}(t) \end{pmatrix} \quad (8)$$

$D_M$  is the dimensionless and used 0.45 ( $< 0.5$ ) for a stable calculation. For  $k = 1$  and  $k = N$  (final mesh point) we have

$$\frac{c_1(t + \Delta t) - c_1(t)}{\Delta t} = \frac{J_0(t) - J_1(t)}{\Delta z}, \quad J_0(t) = -D \frac{c_1(t) - c_0(t)}{\Delta z/2}, \quad J_1(t) = -D \frac{c_2(t) - c_1(t)}{\Delta z}$$

$$\frac{c_N(t + \Delta t) - c_N(t)}{\Delta t} = \frac{J_{N-1}(t) - J_N(t)}{\Delta z}, \quad J_{N-1}(t) = -D \frac{c_N(t) - c_{N-1}(t)}{\Delta z}, \quad J_N(t) = 0$$

$$c_1(t + \Delta t) = c_1(t) + D_M [2c_0(t) - 3c_1(t) + c_2(t)] \quad (9)$$

$$c_N(t + \Delta t) = c_N(t) + D_M [c_{N-1}(t) - c_N(t)] \quad (10)$$

If we use the matrix

$$\begin{pmatrix} c_0(t + \Delta t) \\ c_1(t + \Delta t) \\ c_2(t + \Delta t) \\ \dots \\ \dots \\ c_{N-1}(t + \Delta t) \\ c_N(t + \Delta t) \end{pmatrix} = \begin{pmatrix} ? & ? & 0 & \dots & & & & & & \\ 2D_M & 1 - 3D_M & D_M & 0 & \dots & & & & & \\ 0 & D_M & 1 - 2D_M & D_M & 0 & \dots & & & & \\ \dots & & & & & & & & & \\ \dots & & & & & & & & & \\ 0 & 0 & 0 & \dots & 0 & D_M & 1 - 2D_M & D_M & & \\ 0 & 0 & 0 & \dots & 0 & 0 & D_M & 1 - D_M & & \end{pmatrix} \begin{pmatrix} c_0(t) \\ c_1(t) \\ c_2(t) \\ \dots \\ \dots \\ c_{N-1}(t) \\ c_N(t) \end{pmatrix}$$

We will use

$$E(t) = \begin{cases} E_0 \mp vt, & 0 < t < t_h \\ E_0 \mp 2vt_h \pm vt, & t_h < t < 2t_h \end{cases} \quad (11)$$

$$\Delta E = E_{\max}/i_{\max} = 2vt_h/i_{\max} \quad (12)$$

$$\Delta t = t_{\max}/i_{\max} = 2t_h/i_{\max} \quad (13)$$

$$t(i) = i\Delta t \quad (14)$$

$$\Delta z = \sqrt{\frac{D\Delta t}{D_M}} \quad (15)$$

$$z(k) = k\Delta z \quad (16)$$

$$z_{\max} = 6\sqrt{Dt_h} (> \sqrt{\pi Dt_h}) \quad (17)$$

Example:  $v = 50 \text{ mV s}^{-1}$ ,  $E_{\max} = 2 \times 500 \text{ mV}$ ,  $2t_h = 20 \text{ s}$ ,  $i_{\max} = 10000$ ,  $\Delta t = 2 \times 10^{-3} \text{ s}$ ,  $\Delta E = 0.1 \text{ mV}$ ,  $D_M = 0.45$ ,  $D = 10^{-9} \text{ m}^2 \text{ s}^{-1}$ ,  $\Delta z = 2 \times 10^{-6} \text{ m}$ ,  $z_{\max} = 6 \times 10^{-4} \text{ m}$ ,  $k_{\max} = 300$

I don't like dimensionless simulation, but, it is convenient to use dimensionless concentration unit  $f = c/c^*$ .

$$f_k(t + \Delta t) = f_k(t) + D_M [f_{k+1}(t) - 2f_k(t) + f_{k-1}(t)] \quad (18)$$

Here the position index  $k$ , the time index  $i$ , and the concentration  $f$  and the diffusion constant  $D_M$  are all dimensionless parameters. For the mesh just above the electrode surface ( $k = 1$ )

$$f_1(t + \Delta t) = f_1(t) + D_M [f_2(t) - f_1(t)] - J_{k=0}\Delta t/c^*\Delta z \quad (19)$$

## 2 Initial condition

The initial potential  $E_0$  is chosen for the condition that there is no current  $i = 0$ . This means that the initial potential  $E_0$  is much positive(negative) than the formal redox potential  $E^{\ominus'}$  of for the reduction(oxidation).

$$\text{Reduction : } E_0 > E^{\ominus'} \quad (20)$$

$$c_{\text{Ox}}(z, t = 0) = c^*, \quad c_{\text{Rd}}(z, t = 0) = 0, \quad (21)$$

$$\text{Oxidation : } E_0 < E^{\ominus'} \quad (22)$$

$$c_{\text{Ox}}(z, t = 0) = 0, \quad c_{\text{Rd}}(z, t = 0) = c^* \quad (23)$$

## 3 Boundary condition:Redox at the electrode

In the bulk at  $z = z_{\max}$

$$\text{Reduction : } E_0 > E^{\ominus'} \quad (24)$$

$$c_{\text{Ox}}(z_{\max}, t) = c^*, \quad c_{\text{Rd}}(z_{\max}, t) = 0, \quad (25)$$

$$\text{Oxidation : } E_0 < E^{\ominus'} \quad (26)$$

$$c_{\text{Ox}}(z_{\max}, t) = 0, \quad c_{\text{Rd}}(z_{\max}, t) = c^* \quad (27)$$

Now we consider the redox at the electrode



The diffusion equations for O and R are given

$$\frac{\partial c_{\text{Ox}}}{\partial t} = D_{\text{Ox}} \frac{\partial^2 c_{\text{Ox}}}{\partial z^2} \quad (29)$$

$$\frac{\partial c_{\text{Rd}}}{\partial t} = D_{\text{Rd}} \frac{\partial^2 c_{\text{Rd}}}{\partial z^2} \quad (30)$$

The Faraday current  $i_f$  and  $i_b$  is given by

$$\frac{i_f(t)}{nFA} = -k_f c_{\text{Ox}}(z=0, t) \quad (31)$$

$$\frac{i_b(t)}{nFA} = +k_b c_{\text{Rd}}(z=0, t) \quad (32)$$

The reductive current is defined by negative (e.g. for reductive desorption for  $\text{Au-S-(CH}_2)_n\text{CH}_3 + e^- \rightarrow \text{Au} + (\text{S-(CH}_2)_n\text{CH}_3)$ , then the minus sign for  $i_f(t)$ ). The dimension of the lhs of the above equations are the same as flux  $[\text{A} / (\text{C mol}^{-1} \text{m}^2)] = [\text{C s}^{-1} / (\text{C mol}^{-1} \text{m}^2)] = [\text{mol m}^{-2} \text{s}^{-1}]$ , then the dimension of  $k_f$  and  $k_b$  is  $[\text{m s}^{-1}]$ .

The net current density  $i$  becomes

$$\frac{i}{nFA} = \frac{i_f + i_b}{nFA} = -k_f c_{\text{Ox}}(z=0, t) + k_b c_{\text{Rd}}(z=0, t) \quad (33)$$

$$(34)$$

If we use the Butler-(Erdey-Grüz)-Volmer relation, we can get

$$k_f(t) = k_0 \exp\left[-\frac{\alpha nF}{RT}(E(t) - E^{\ominus'})\right] \quad (35)$$

$$k_b(t) = k_0 \exp\left[\frac{(1-\alpha)nF}{RT}(E(t) - E^{\ominus'})\right] \quad (36)$$

In the case that  $k_0$  is great enough, the system becomes "reversible".

The change of the concentrations O and R at the electrode may be given by

$$\frac{\partial c_{\text{Ox}}(z=0, t)}{\partial t} dV = [-J_{\text{Ox}}(z=0, t) - k_f(t)c_{\text{Ox}}(z=0, t) + k_b(t)c_{\text{Rd}}(z=0, t)] dS \quad (37)$$

$$\frac{\partial c_{\text{Rd}}(z=0, t)}{\partial t} dV = [-J_{\text{Rd}}(z=0, t) - k_b(t)c_{\text{Rd}}(z=0, t) + k_f(t)c_{\text{Ox}}(z=0, t)] dS \quad (38)$$

$$dV = \Delta x \Delta y \Delta z_{\text{surface}}, \quad dS = \Delta x \Delta y \quad (39)$$

Hereafter we use the definition for  $c_{\text{Ox}}(z, t)$  and  $c_{\text{Rd}}(z, t)$

$$c_{\text{Ox}}(z, t) \equiv c_{\text{Ox}}(i, t), \quad c_{\text{Rd}}(z, t) \equiv c_{\text{Rd}}(i, t); \quad z(i) = z \quad (40)$$

The discretized form of the above equations are

$$J_{\text{Ox}}(z=0, t) = -\frac{2D_{\text{Ox}}}{\Delta z} [c_{\text{Ox}}(1, t) - c_{\text{Ox}}(0, t)] \quad (41)$$

$$J_{\text{Rd}}(z=0, t) = -\frac{2D_{\text{Rd}}}{\Delta z} [c_{\text{Rd}}(1, t) - c_{\text{Rd}}(0, t)] \quad (42)$$

$$\frac{c_{\text{Ox}}(0, t + \Delta t) - c_{\text{Ox}}(0, t)}{\Delta t} \Delta z_{\text{surface}} = \frac{2D_{\text{Ox}}}{\Delta z} [c_{\text{Ox}}(1, t) - c_{\text{Ox}}(0, t)] - k_f(t)c_{\text{Ox}}(0, t) + k_b(t)c_{\text{Rd}}(0, t) \quad (43)$$

$$\frac{c_{\text{Rd}}(0, t + \Delta t) - c_{\text{Rd}}(0, t)}{\Delta t} \Delta z_{\text{surface}} = \frac{2D_{\text{Rd}}}{\Delta z} [c_{\text{Rd}}(1, t) - c_{\text{Rd}}(0, t)] - k_b(t)c_{\text{Rd}}(0, t) + k_f(t)c_{\text{Ox}}(0, t) \quad (44)$$

In the limit  $\Delta z_{\text{surface}} \rightarrow 0$ , the lhs of Eq.(43) and Eq.(44) become zero, then from Eq.(43) + Eq.(44)

$$D_{\text{Ox}}[c_{\text{Ox}}(1, t) - c_{\text{Ox}}(0, t)] = -D_{\text{Rd}}[c_{\text{Rd}}(1, t) - c_{\text{Rd}}(0, t)] \quad (45)$$

This means that the fluxes of Ox and Red are cancelled out. From Eq.(43) and Eq.(44)

$$\left[ \frac{2D_{\text{Ox}}}{\Delta z} + k_f(t) \right] c_{\text{Ox}}(0, t) = \frac{2D_{\text{Ox}}}{\Delta z} c_{\text{Ox}}(1, t) + k_b(t)c_{\text{Rd}}(0, t) \quad (46)$$

$$c_{\text{Rd}}(0, t) = \frac{\frac{2D_{\text{Rd}}}{\Delta z} c_{\text{Rd}}(1, t) + k_f(t)c_{\text{Ox}}(0, t)}{\frac{2D_{\text{Rd}}}{\Delta z} + k_b(t)} \quad (47)$$

$$\left[ \frac{2D_{\text{Ox}}}{\Delta z} + k_f(t) \right] c_{\text{Ox}}(0, t) = \frac{2D_{\text{Ox}}}{\Delta z} c_{\text{Ox}}(1, t) + k_b(t) \frac{\frac{2D_{\text{Rd}}}{\Delta z} c_{\text{Rd}}(1, t) + k_f(t) c_{\text{Ox}}(0, t)}{\frac{2D_{\text{Rd}}}{\Delta z} + k_b(t)} \quad (48)$$

$$\left[ \frac{2D_{\text{Ox}}}{\Delta z} + k_f(t) - \frac{k_f(t)k_b(t)}{\frac{2D_{\text{Rd}}}{\Delta z} + k_b(t)} \right] c_{\text{Ox}}(0, t) = \frac{2D_{\text{Ox}}}{\Delta z} c_{\text{Ox}}(1, t) + \frac{k_b(t) \frac{2D_{\text{Rd}}}{\Delta z}}{\frac{2D_{\text{Rd}}}{\Delta z} + k_b(t)} c_{\text{Rd}}(1, t) \quad (49)$$

$$c_{\text{Ox}}(0, t) = B_1 c_{\text{Ox}}(1, t) + B_2 c_{\text{Rd}}(1, t) \quad (50)$$

$$B_1 = \frac{\frac{2D_{\text{Ox}}}{\Delta z}}{\frac{2D_{\text{Ox}}}{\Delta z} + k_f(t) - \frac{k_f(t)k_b(t)}{\frac{2D_{\text{Rd}}}{\Delta z} + k_b(t)}} \quad (51)$$

$$B_2 = \frac{\frac{k_b(t) \frac{2D_{\text{Rd}}}{\Delta z}}{\frac{2D_{\text{Rd}}}{\Delta z} + k_b(t)}}{\frac{2D_{\text{Ox}}}{\Delta z} + k_f(t) - \frac{k_f(t)k_b(t)}{\frac{2D_{\text{Rd}}}{\Delta z} + k_b(t)}} \quad (52)$$

$$\left[ \frac{2D_{\text{Rd}}}{\Delta z} + k_b(t) \right] c_{\text{Rd}}(0, t) = \frac{2D_{\text{Rd}}}{\Delta z} c_{\text{Rd}}(1, t) + k_f(t) c_{\text{Ox}}(0, t) \quad (53)$$

$$= \frac{2D_{\text{Rd}}}{\Delta z} c_{\text{Rd}}(1, t) + k_f(t) [B_1 c_{\text{Ox}}(1, t) + B_2 c_{\text{Rd}}(1, t)] \quad (54)$$

$$c_{\text{Rd}}(0, t) = \frac{\frac{2D_{\text{Rd}}}{\Delta z} + k_f(t) B_2}{\frac{2D_{\text{Rd}}}{\Delta z} + k_b(t)} c_{\text{Rd}}(1, t) + \frac{k_f(t) B_1}{\frac{2D_{\text{Rd}}}{\Delta z} + k_b(t)} c_{\text{Ox}}(1, t) \quad (55)$$

Using Eq.(50, 51, 52, 55)  $c_{\text{Ox}}(0, t)$  and  $c_{\text{Rd}}(0, t)$  can be determined from  $c_{\text{Ox}}(1, t)$ ,  $c_{\text{Rd}}(1, t)$ ,  $k_f(t)$ ,  $k_b(t)$ .

If we use the matrix and  $D_M^{\text{Ox}} = (D_{\text{Ox}}\Delta t)/(\Delta z)^2$ ,  $D_M^{\text{Rd}} = (D_{\text{Rd}}\Delta t)/(\Delta z)^2$

$$\begin{pmatrix} c_{\text{Ox}}(1, t + \Delta t) - 2D_M^{\text{Ox}} c_{\text{Ox}}(0, t) \\ c_{\text{Ox}}(2, t + \Delta t) \\ c_{\text{Ox}}(3, t + \Delta t) \\ \dots \\ \dots \\ c_{\text{Ox}}(N-1, t + \Delta t) \\ c_{\text{Ox}}(N, t + \Delta t) \end{pmatrix} = \begin{pmatrix} 1 - 3D_M^{\text{Ox}} & D_M^{\text{Ox}} & 0 & \dots \\ D_M^{\text{Ox}} & 1 - 2D_M^{\text{Ox}} & D_M^{\text{Ox}} & 0 & \dots \\ 0 & D_M^{\text{Ox}} & 1 - 2D_M^{\text{Ox}} & D_M^{\text{Ox}} & 0 & \dots \\ \dots & \dots & \dots & \dots & \dots & \dots \\ \dots & \dots & \dots & \dots & \dots & \dots \\ 0 & 0 & 0 & \dots & 0 & D_M^{\text{Ox}} & 1 - 2D_M^{\text{Ox}} & D_M^{\text{Ox}} \\ 0 & 0 & 0 & \dots & 0 & 0 & D_M^{\text{Ox}} & 1 - D_M^{\text{Ox}} \end{pmatrix} \begin{pmatrix} c_{\text{Ox}}(1, t) \\ c_{\text{Ox}}(2, t) \\ c_{\text{Ox}}(3, t) \\ \dots \\ \dots \\ c_{\text{Ox}}(N-1, t) \\ c_{\text{Ox}}(N, t) \end{pmatrix} \quad (56)$$

$$\begin{pmatrix} c_{\text{Rd}}(1, t + \Delta t) - 2D_M^{\text{Rd}} c_{\text{Rd}}(0, t) \\ c_{\text{Rd}}(2, t + \Delta t) \\ c_{\text{Rd}}(3, t + \Delta t) \\ \dots \\ \dots \\ c_{\text{Rd}}(N-1, t + \Delta t) \\ c_{\text{Rd}}(N, t + \Delta t) \end{pmatrix} = \begin{pmatrix} 1 - 3D_M^{\text{Rd}} & D_M^{\text{Rd}} & 0 & \dots \\ D_M^{\text{Rd}} & 1 - 2D_M^{\text{Rd}} & D_M^{\text{Rd}} & 0 & \dots \\ 0 & D_M^{\text{Rd}} & 1 - 2D_M^{\text{Rd}} & D_M^{\text{Rd}} & 0 & \dots \\ \dots & \dots & \dots & \dots & \dots & \dots \\ \dots & \dots & \dots & \dots & \dots & \dots \\ 0 & 0 & 0 & \dots & 0 & D_M^{\text{Rd}} & 1 - 2D_M^{\text{Rd}} & D_M^{\text{Rd}} \\ 0 & 0 & 0 & \dots & 0 & 0 & D_M^{\text{Rd}} & 1 - D_M^{\text{Rd}} \end{pmatrix} \begin{pmatrix} c_{\text{Rd}}(1, t) \\ c_{\text{Rd}}(2, t) \\ c_{\text{Rd}}(3, t) \\ \dots \\ \dots \\ c_{\text{Rd}}(N-1, t) \\ c_{\text{Rd}}(N, t) \end{pmatrix} \quad (57)$$

If the concentration at  $z = z_{\text{max}}$  is always the same, this means

$$c_{\text{Ox}}(N-1, t) = c_{\text{Ox}}(N, t) \quad (58)$$

$$c_{\text{Rd}}(N-1, t) = c_{\text{Rd}}(N, t) \quad (59)$$

## 4 Fortran programs and some examples

You don't need to buy the CV simulation software. You can write the program. In the following we show the **125-lines-only Fortran program** to simulate Butler-Volmer type CVs.

```
1 c23456789| Source code written by M. Yamamoto EHCC Kyoto-U.
2 c      CV simulation :unit MKS
3 c      Butler-Volmer type :reversible q-rev. irrev CVs can simulate
4      implicit real*8 (a-h,o-z)
5      double precision :: cox(0:10000),crd(0:10000),z(0:10000)
6      double precision :: cox_new(0:10000),crd_new(0:10000)
7 c----- input parameter here
8      dox=1.0d-9 ! diffusion const. of Ox  m2 s-1
9      drd=1.0d-9 ! diffusion const. of Rd  m2 s-1
10     e0=0.25d0  ! start potential V
11     v=-0.05d0 ! sweep rate    V s-1;  v < 0:redreuction
12     th=10.0d0 ! half time of sweep
13     em=e0+v*th ! E(t)=e0 + v*t      0<= t <= th,
14     ef=e0+2.0d0*v*th-2.0d0*v*th ! E(t)= e0+2*v*th-v*t  th <= t <= 2*th
15     eoXrd=0.0d0 ! formal potential of Ox-Red : V
16     imax=20000 ! time mesh
17     emax=2.0d0*(e0-em) ! full scale of E:V
18     tmax=2.0d0*th ! time to sweep
19     delta_t=tmax/real(imax) ! time step in s
20     dm=0.45d0 ! step parameter usually taken as 0.45
21     dmin=min(dox,drd) ! smaller diffusion constant
22     dmax=max(dox,drd) ! greater diffusion constant
23     delta_z=sqrt(dox*delta_t/dm) ! mesh in z coord. in m
24     zmax=6.0d0*sqrt(dmax*th) ! max in z : most outer plane in solution
25     kmax=int(zmax/delta_z)+1 ! # of mesh in z
26     cox0=0.01d0*1000.0d0 ! c_Ox in bulk / mol m-3 be careful 1 M = 1 mol dm-3 = 1000 mol m-3
27     crd0=0.0d0 ! c_Rd in bulk / mol m-3
28     ek0=1.0d-5 !k0 / m s-1
29     alpha=0.5d0 !alpha parameter in Butlet-Volmer Eq.
30     temp=300.0d0 !temp / K
31     en=1.0d0 !n: Ox(z) + ne = Rd(z-n) number of electron
32 c----- print parameters
33     write (6,*) 'dox: diff. const. of Ox in m2 s-1',dox
34     write (6,*) 'drd: diff. const. of Rd in m2 s-1',drd
35     write (6,*) 'e0: start potential in V',e0
36     write (6,*) 'v:sweep rate in V s-1; v < 0:redreuction', v
37     write (6,*) 'th: in s; E(t)=e0 + v*t 0<= t <= th',th
38     write (6,*) 'em=e0+v*th, ef=e0',em,ef
39     write (6,*) 'eoXrd: formal potential of Ox-Red in V',eoXrd
40     write (6,*) 'imax: # of time mesh',imax
41     write (6,*) 'emax: full scale of E in V',emax
42     write (6,*) 'tmax: time to sweep in s',tmax
43     write (6,*) 'delta_t: time step in s', delta_t
44     write (6,*) 'dm: step parameter usually taken as 0.45', dm
45     write (6,*) 'dmin=min(dox,drd),dmax=max(dox,drd)',dmin,dmax
46     write (6,*) 'delta_z:mesh in z coord. in m',delta_z
47     write (6,*) 'zmax: max in z / m ',zmax
48     write (6,*) 'kmax: # of mesh in z',kmax
49     write (6,*) 'cox0: 1000*c_Ox / M in bulk ',cox0
50     write (6,*) 'crd0: 1000*c_Rd / M in bulk ',crd0
51     write (6,*) 'ek0: rate const. k0 / m s-1',ek0
52     write (6,*) 'alpha:alpha parameter in Butlet-Volmer Eq.',alpha
53     write (6,*) 'temp:temp / K',temp
54     write (6,*) 'en: n: Ox(z) + ne = Rd(z-n)', en
55 c ----- physical const.
56     f=96485.3399d0 ! Faraday const.
```

```

57      r=8.314472d0      ! molar gas const.
58 c    ----- diff coeff. without dimension
59      dmox=dox*delta_t/delta_z/delta_z
60      dmrdr=drd*delta_t/delta_z/delta_z
61      write (6,*) 'dmox,dmrdr',dmox,dmrdr
62 c    ---- z coordinate
63      do k=0, kmax
64          z(k)=real(k)*delta_z
65      enddo
66 c    ---- initial condition ---
67      do k=1, kmax
68          cox(k)=cox0
69          crd(k)=crd0
70      enddo
71 c    ----- surface concentration
72      et=e0
73      ekft=ek0*exp(-alpha*en*f/r/temp*(et-eoxrd))
74      ekbt=ek0*exp((1.0d0-alpha)*en*f/r/temp*(et-eoxrd))
75      b1=2.0d0*dox/delta_z/(2.0d0*dox/delta_z+ekft-
76 &      ekft*ekbt/(2.0d0*drd/delta_z+ekbt))
77      b2=ekbt*2.0d0*drd/delta_z/(2.0d0*drd/delta_z+ekbt)/
78 &      (2.0d0*dox/delta_z+ekft-ekft*ekbt/(2.0d0*drd/delta_z+ekbt))
79      cox(0)=b1*cox(1)+b2*crd(1)
80      crd(0)=(2.0d0*drd/delta_z+ekft*b2)/(2.0d0*drd/delta_z+ekbt)
81 &      *crd(1)+ekft*b1/(2.0d0*drd/delta_z+ekbt)*cox(1)
82      write (6,*) 'b1,b2,cox(0),crd(0)',b1,b2,cox(0),crd(0)
83      write (6,*) 'time, potential / V, current density / Am-2,
84 &      cox(surface), crd(surface),cox(1),crd(1),ekft,ekbt'
85 c    ----- time routine i=1, imax
86      do i=1, imax
87          time=delta_t*real(i)
88          if (time <= th ) then
89              et=e0+v*time
90          else
91              et=e0+2.0d0*v*th-v*time
92          endif
93          ekft=ek0*exp(-alpha*en*f/r/temp*(et-eoxrd))
94          ekbt=ek0*exp((1.0d0-alpha)*en*f/r/temp*(et-eoxrd))
95          cox_new(1)=2.0d0*dmox*cox(0)+(1.0d0-3.0d0*dmox)*cox(1)
96 &          +dmox*cox(2)
97          crd_new(1)=2.0d0*dmrdr*crd(0)+(1.0d0-3.0d0*dmrdr)*crd(1)
98 &          +dmrdr*crd(2)
99          do k=2, kmax-1
100             cox_new(k)=dmox*cox(k-1)+(1.0d0-2.0d0*dmox)*cox(k)
101 &             +dmox*cox(k+1)
102             crd_new(k)=dmrdr*crd(k-1)+(1.0d0-2.0d0*dmrdr)*crd(k)
103 &             +dmrdr*crd(k+1)
104         enddo
105
106         b1=2.0d0*dox/delta_z/(2.0d0*dox/delta_z+ekft-
107 &         ekft*ekbt/(2.0d0*drd/delta_z+ekbt))
108         b2=ekbt*2.0d0*drd/delta_z/(2.0d0*drd/delta_z+ekbt)/
109 &         (2.0d0*dox/delta_z+ekft-ekft*ekbt/(2.0d0*drd/delta_z+ekbt))
110
111         cox_new(0)=b1*cox_new(1)+b2*crd_new(1)
112         crd_new(0)=(2.0d0*drd/delta_z+ekft*b2)/(2.0d0*drd/delta_z+ekbt)
113 &         *crd_new(1)+ekft*b1/(2.0d0*drd/delta_z+ekbt)*cox_new(1)
114
115         cox_new(kmax)=cox0
116         crd_new(kmax)=crd0
117         current_dens=en*f*(-ekft*cox_new(0)+ekbt*crd_new(0))

```

```

118     write (6,'(9e15.7)') time,et,current_dens,cox_new(0),crd_new(0)
119     &                                     ,cox_new(1),crd_new(1),ekft,ekbt
120     do k=0, kmax
121         cox(k)=cox_new(k)
122         crd(k)=crd_new(k)
123     enddo
124
125     enddo ! i loop
126     stop
127     end

```

Simulation is done by the following condition. dox: diff. const. of Ox in m<sup>2</sup> s<sup>-1</sup> 1.0E-009, drd: diff. const. of Rd in m<sup>2</sup> s<sup>-1</sup> 1.0E-009, e0: start potential in V 0.25, v:sweep rate in V s<sup>-1</sup>;  $v < 0$  :reduction -5.0E-002, th: in s;  $E(t) = e0 + v * t0 \leq t \leq th$  10.0, em=e0+v\*th, ef=e0 -0.25, 0.250, eoXrd: formal potential of Ox-Red in V 0.0, imax: number of time mesh 20000, emax: full scale of E in V 1.0, tmax: time to sweep in s 20.0, delta t: time step in s 1.0E-003, dm: step parameter usually taken as 0.45, delta z:mesh in z coor. in m 1.4907E-006, zmax: max in z / m 6.0E-004, kmax: number of mesh in z 403, cox0: 1000\*c Ox in bulk / M 10.0, crd0: 1000\*c Rd in bulk / M 0.0, ek0: rate const. k<sub>0</sub> / m s<sup>-1</sup> form 100 to 1.0E-007, alpha:alpha parameter in Butler-Volmer Eq. 0.5, temp:temp / K 300.0, en: n: Ox(z) + ne = Rd(z-n) 1.0.




Figure 2: Simulation of CVs for Butler-Volmer rate processes

We checked the program by Nicholson's plot  $\Psi$  and peak separation [Anal. Chem. 37, 1351-1355(1965)], where  $\Psi$  is given by

$$\Psi = \frac{(D_{\text{Ox}}/D_{\text{Rd}})^{\alpha/2} k_0^{\text{app}}}{\sqrt{\pi D_{\text{Ox}} n F v / (RT)}} \quad (60)$$

As shown in the figures below, the program may work well.

Please remember that when  $k_0 \geq 10^{-2} \text{ m s}^{-1} = 1 \text{ cm s}^{-1}$ , the CV looks like the reversible one under the experimental conditions shown above.

**We also checked the magnitude of the current density is OK!**




Figure 3: Nicholson plot

## 5 Adsorption (not finished)

Just consider the coverage  $\theta_{\text{Ox}}$  and  $\theta_{\text{Rd}}$  and the local equilibrium holds. The adsorption(desorption) equilibrium can be written in Langmuir, mean-field, and quasi-chemical approximation. In the Langmuir approximation we obtain

$$k_{\text{ad}}^{\text{Ox}} c_{\text{Ox}}(0, t)(1 - \theta_{\text{Ox}}) = k_{\text{de}}^{\text{Ox}} \theta_{\text{Ox}}, \quad K^{\text{Ox}} c_{\text{Ox}}(0, t) = \frac{\theta_{\text{Ox}}}{1 - \theta_{\text{Ox}}} \quad (61)$$

$$k_{\text{ad}}^{\text{Rd}} c_{\text{Rd}}(0, t)(1 - \theta_{\text{Rd}}) = k_{\text{de}}^{\text{Rd}} \theta_{\text{Rd}}, \quad K^{\text{Rd}} c_{\text{Rd}}(0, t) = \frac{\theta_{\text{Rd}}}{1 - \theta_{\text{Rd}}} \quad (62)$$

If we assume that the redox reaction takes place at the adsorption site,

$$\frac{i}{nFA} = i_f + i_b = -k_f \theta_{\text{Ox}}(t) + k_b \theta_{\text{Rd}}(t) \quad (63)$$

$$(64)$$

Which species can take part in the redox reaction?






Figure 4: Nicholson plot




Figure 5: dependence of the peak separation on the switching potential for the reversible redox reaction. Simulation is done by the following condition. dox: diff. const. of Ox in m<sup>2</sup> s<sup>-1</sup> 1.0E-009, drd: diff. const. of Rd in m<sup>2</sup> s<sup>-1</sup> 1.0E-009, e0: start potential in V 0.25, v:sweep rate in V s<sup>-1</sup>;  $v < 0$  :reduction -5.0E-002, eoxrd: formal potential of Ox-Red in V 0.0, imax: number of time mesh 20000, delta t: time step in s 1.0E-003, dm: step parameter usually taken as 0.45, delta z:mesh in z coord. in m 1.4907E-006, zmax: max in z / m 6.0E-004, kmax: number of mesh in z 403, cox0: 1000\*c Ox in bulk / M 10.0, crd0: 1000\*c Rd in bulk / M 0.0, ek0: rate const. k0 / m s<sup>-1</sup> 100, alpha:alpha parameter in Butlet-Volmer Eq. 0.5, temp:temp / K 300.0, en: n: Ox(z) + ne = Rd(z-n) 1.0.